
1) Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 13.3.2.2, which is part of the LTE test suite. 

The test case can be demonstrated to run with one LTE UE. Execution log files are provided as evidence. 

3) Verification Test Summary

Test Case:
13.3.2.2
Test Group:
Multi layer Procedures
ATS Version:
iwd-EUTRA-B2012-03_D12wk24
System Simulator used:
Anritsu Protocol Conformance Test System ME7832L
UE used:
Nvidia Icera 410
Verification Status:
PASS
4) Corrections required for test case 13.3.2.2
4.1 Introduction

The below changes were required to make test case 13.3.2.2 run correctly with 1 LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-B2012-03_D12wk24 release.

4.2 Change 1- Correction to testcase TC_13_3_2_2
	Test Case name
	testcase TC_13_3_2_2

	Reason for change
	Insufficient test case over all guard timer value. (That is,  marginal in this case)


	Summary of change
	Change guard timer from 6minutes to 7 minutes

	TTCN module
	LTE_EPS_TS_Testcases.ttcn


Before change

	  testcase TC_13_3_2_2() runs on MTC_LTE system SYSTEM_LTE {

    // @purpose

    //   Inter-system connection re-establishment / E-UTRAN to GPRS / Further data are to be transferred

    var EUTRA_PTC    v_EUTRA      := null;

    var UTRAN_PTC    v_UTRAN      := null;

    :

    :  

    timer t_GuardTimer := int2float(360);
    v_EUTRA := EUTRA_PTC.create alive;

    v_GERAN := GERAN_PTC.create alive;

    :

    :  

  }


After change

	  testcase TC_13_3_2_2() runs on MTC_LTE system SYSTEM_LTE {

    // @purpose

    //   Inter-system connection re-establishment / E-UTRAN to GPRS / Further data are to be transferred

    var EUTRA_PTC    v_EUTRA      := null;

    var UTRAN_PTC    v_UTRAN      := null;

    :

    :  

    timer t_GuardTimer := int2float(420);
    v_EUTRA := EUTRA_PTC.create alive;

    v_GERAN := GERAN_PTC.create alive;

    :

    :  

  }


4.3 Change 2 - Correction to function f_TC_13_3_2_2_GERAN

	Test Case name
	f_TC_13_3_2_2_GERAN

	Reason for change
	Current test case uses function f_GPRS_ReceiveUplinkData to receive uplink packet data PDU. But, this function is only designed to receive one phase uplink signalling PDU and therefore can not be used to receive U-plane packet data as the uplink packet data must be received using two phase access instead.

	Summary of change
	Added new function f_GPRS_ReceiveUplinkPktData to receive uplink packet data as appropriate. A new two access function (See change 2.1)

	TTCN module
	MultiLayer_Procedures_GERAN.ttcn


Before change

	  function f_TC_13_3_2_2_GERAN () runs on GERAN_PTC

  {

      /* Inter-system connection re-establishment / E-UTRAN to GPRS / Further data are to be transferred */

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var B3_Type v_UpdateType;

    var PDP_ContextStatus v_PDP_Context_Status;

    var GERAN_SecurityInfo_Type v_SecurityInfo;

    var octetstring v_PLMN;

    var O1_Type v_RAC;

    var O2_Type v_LAC;

    :

    :

    //@siclog "Step 15" siclog@

    //Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context? */

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24,

                                             car_G_LLC_UnitData_IND(cr_PacketData (crs_IPv4_AnyIcmpEchoReply_1 ( px_IPv4_Address1_UE, px_IPv4_Address1_UE)),

                                             tsc_LLCSapi_LoopBack));

    f_GERAN_PreliminaryPass(__FILE__, __LINE__, "Test case 13.3.2.2 Step 15");

    //Tell to EUTRA PTC Test case is now finished on GERAN side

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_TestBody_Set(false);

    //Switch/power off UE and release cell

    f_GERAN_Postamble (geran_Cell24, G1_IDLE);

  }


After change

	  function f_TC_13_3_2_2_GERAN () runs on GERAN_PTC

  {

      /* Inter-system connection re-establishment / E-UTRAN to GPRS / Further data are to be transferred */

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var B3_Type v_UpdateType;

    var PDP_ContextStatus v_PDP_Context_Status;

    var GERAN_SecurityInfo_Type v_SecurityInfo;

    var octetstring v_PLMN;

    var O1_Type v_RAC;

    var O2_Type v_LAC;

    :

    :

    //@siclog "Step 15" siclog@

    //Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context? */

    v_LLCDataInd := f_GPRS_ReceiveUplinkPktData (geran_Cell24,

                                             car_G_LLC_UnitData_IND(cr_PacketData (crs_IPv4_AnyIcmpEchoReply_1 ( px_IPv4_Address1_UE, px_IPv4_Address1_UE)),

                                             tsc_LLCSapi_LoopBack));

    f_GERAN_PreliminaryPass(__FILE__, __LINE__, "Test case 13.3.2.2 Step 15");

    //Tell to EUTRA PTC Test case is now finished on GERAN side

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_TestBody_Set(false);

    //Switch/power off UE and release cell

    f_GERAN_Postamble (geran_Cell24, G1_IDLE);

  }


4.3.1 Change 2.1 – Add new function f_GPRS_ReceiveUplinkPktData
	Function name
	f_GPRS_ReceiveUplinkPktData

	Reason for change
	Current test case uses function f_GPRS_ReceiveUplinkData to receive uplink packet data PDU. But, this function is only designed to receive one phase uplink signalling PDU and therefore can not be used to receive U-plane packet data as the uplink packet data must be received using two phase access instead.

	Summary of change
	Added new function f_GPRS_ReceiveUplinkPktData to receive uplink packet data as appropriate. A new two access function (See change 2.1.1)

	TTCN module
	CommonGERAN\GERAN_CommonFunctions.ttcn

	MCC160 Comment
	


New Template
	// Anritsu_13322 - new function to handle 2 phase access

function f_GPRS_ReceiveUplinkPktData (GERAN_CellId_Type p_CellId,  

                                  template G_LLC_DATAMESSAGE_IND p_DataMsgInd,  

                                  charstring p_StepNo := "") runs on GERAN_PTC  return G_LLC_DATAMESSAGE_IND  

{  

  var G_L2_DATAMESSAGE_IND v_ChanReq;  

  var template (omit) G_LLC_DATAMESSAGE_IND v_ULDataInd;  

  G_L2.receive (car_G_L2_ACCESS_IND ( p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqTwoPhase )) -> value v_ChanReq;  

  if (p_StepNo != "") {  

    f_GERAN_PreliminaryPass (__FILE__, __LINE__, p_StepNo);  

  }  

  v_ULDataInd := fl_ULTBFTwoPhase (p_CellId, v_ChanReq.access, p_DataMsgInd); // @sic R5s120050 MCC160 implementation sic@  

  return valueof(v_ULDataInd); // return this to the calling function  

}  


4.3.1.1 Change 2.1.1 – Addition of new function fl_ULTBFTwoPhase
	Function name
	fl_ULTBFTwoPhase














	Reason for change
	Create new function handle two phase access.

	Summary of change
	Summary of the function is listed below.

1. Receive single block packet access CHANNEL REQUEST PDU

2.  Send Immediate Assignment PDU with single block allocation (See change 2.1.1.1 for message/IEdetail)
3. Receive Packet Resource Request PDU

4. Send Packet Uplink Assignment

5. Received “looped backed” uplink packet data.  (Ref 2.1.1.2 message/IE detail)

	TTCN module
	CommonGERAN\GERAN_CommonFunctions.ttcn

	MCC160 Comment
	


New Template
	function fl_ULTBFTwoPhase(GERAN_CellId_Type p_CellId,  

                          G_L2_ACCESS_IND p_AccessInd,  

                          template G_LLC_DATAMESSAGE_IND p_DataMsgInd := omit) runs on GERAN_PTC return template (omit) G_LLC_DATAMESSAGE_IND  

{   

  var G_LLC_DATAMESSAGE_IND v_ULDataInd;  

  var G_CPHY_CONFIG_CNF v_G_CL1_ComingFN_CNF;  

  var G_RLC_DATAMESSAGE_IND v_G_RLC_ControlMsg_IND;  

  var RFN v_StartingTime;  

  var TLLI v_NewTLLI, v_TLLI;  

  var B5_Type v_Gamma := f_GERAN_GAMMA_Get(p_CellId);  

  var B10_Type v_GPRSFreq := f_GERAN_GPRSFreq_Get(p_CellId);  

  var integer v_RR_RA;  

  var RFN v_RFN;  

  var B10_Type v_GPRS_ARFCN := f_GERAN_GPRSFreq_Get (p_CellId);  

  f_GERAN_RLC_CONFIG_Common(cas_ActivateTBF_UplinkDyn(p_CellId, 1));  // Configure the Uplink TBF for Dynamic Allocation, USF 0 on timeslot 4      //Anritsu TTCN WA for GERAN  

  v_RFN := p_AccessInd.rfn;  

  v_RR_RA := bit2int(p_AccessInd.burst.chReq.estCauRandomRef);  

  v_G_CL1_ComingFN_CNF := f_GERAN_CPHY_CONFIG_ComingFN(cas_G_CL1_ComingFN_REQ( p_CellId, tsc_PhyCh1, tsc_PACCH_F));  

  v_StartingTime := v_G_CL1_ComingFN_CNF.comingFN;  

  // Send IA  

  G_L2.send (cas_G_L2_UNITDATA_REQ (p_CellId,  

                                    tsc_PhyCh0,  

                                    tsc_AGCH, 15,  

                                    cs_G_RFN_Omit,  

                                    cas_ImmAss (cs_P_ImmediateAssignment ('0'B,  

                                                                          v_GPRSFreq,  

                                                                          v_RR_RA,  

                                                                          v_RFN,  

                                                                          cs_IARO_PUA_SingleBlock( v_Gamma,  

                                                                                      (v_StartingTime.t1_ & v_StartingTime.t3 & v_StartingTime.t2))))));  // @sic GP-110839 sic@  

  // Configure the Uplink TBF for Dynamic Allocation, USF 0 on timeslot 4  

  f_GERAN_RLC_CONFIG_Common(cas_ActivateTBF_UplinkDyn(p_CellId, 1), dontWaitForCNF);  // don't wait for CNF to avoid race condition with incoming car_G_RLC_ControlMsg_IND  

  // Now wait for PRR      

  alt {  

    [] G_RLC.receive (car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketResourceRequestAny))  

      -> value v_G_RLC_ControlMsg_IND  

      {  

        v_NewTLLI := v_G_RLC_ControlMsg_IND.ctrl.msg.pktResReq.tlli;  

        v_TLLI := f_GERAN_TLLI_Get(p_CellId);  

        if (v_NewTLLI != v_TLLI)  //if the TLLI hasn't changed, don't bother calling LLCAssignReq  

          {//Assign the value received from the MS  

            f_LLC_TLLI_Assign (p_CellId, v_TLLI, v_NewTLLI, dontWaitForCNF);  // don't wait for CNF to avoid race condition with incoming data (L3 message)  

            f_GERAN_OldTLLI_Set(p_CellId, v_TLLI);  

            f_GERAN_TLLI_Set(p_CellId, v_NewTLLI);  

          }  

      }  

  }  

  // Now send PUA  

  G_RLC.send (cas_G_RLC_ControlMsg_REQ_PACCH( p_CellId, tsc_PhyCh1, 1, tsc_UplinkTFI, cs_PacketUplinkAssignment_13322(v_GPRS_ARFCN, v_TLLI)));  

  if (ispresent(p_DataMsgInd)) {  

    interleave {  // to resolve RACE CONDITION between G_LLC and G_RLC  

      [] G_LLC.receive (p_DataMsgInd) -> value v_ULDataInd {}                                                // Receive the message  

      [] G_RLC.receive (car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement)) {}  // receive PCA that's sent when the TBF is released  

    }  

    return v_ULDataInd;  

  }  

  return omit;  

}  


4.3.1.1.1 Change 2.1.1.1 – Add new template IARO cs_IARO_PUA_SingleBlock
	Template name
	cs_IARO_PUA_SingleBlock

	Reason for change
	Require new template to contain single black uplink assigment

	Summary of change
	Create new template to contain single black uplink assignment. See change 2.1.1.1.1 for message/IE detail.

	TTCN module
	CommonGERAN\GERAN_Templates.ttcn

	MCC160 Comment
	


New Template
	// Anritsu_13322 - add new template for single block assignment

template (value) IARO cs_IARO_PUA_SingleBlock(B5_Type p_Gamma,  

                                  B16_Type p_StartTime) :=  

{ /* @status    APPROVED */  

  constructMask   := '11'B,  

  compressInterRATHO := omit,                      // present if constructMask = LL  

  reservMask1     := omit,                         // present if constructMask = LH  

  reservMask2     := omit,                         // present if reservMask1 = 0  

  eGPRS_UplinkAssign := omit,                      // not present if reservMask2 = 1 or reservMask1 = 1 or constructMask <> LH  

  multiBlocksPDA  := omit,                         // not present if reservMask2 = 0 or reservMask1 = 1 or constructMask <> LH  

  lengthOfFreqParam := omit,                       // present if constructMask = HL  

  freqParam_bt    := omit,                         // frequency parameters before time  

                                                   // not present if lengthOfFreqParam = 0  

  compressInterRATHO1_Type := omit,                     // present if constructMask = HL  

  mask1           := '0'B,                         // present if constructMask = HH  

  mask2           := '0'B,  

  packetUplinkAssign := cs_PktUplinkAss_IARO_SingleBlock(p_Gamma, p_StartTime), // not present if mask1 = 1 or mask2 = 1 or constructMask <> HH  

  packetDownlinkAssign := omit,                    // not present if mask1 = 1 or mask2 = 0 or constructMask <> HH  

  secondPacketAssign := omit,                       // present if mask1 = 1 and constructMask = HH  

  r10Mask         := '0'B,                       // @sic R5s110178 Baseline Moving sic@  

  implicitRejectCS := omit,    // not present if r10Mask = L @sic R5s110178 Baseline Moving sic@  

  implicitRejectPS := omit    // not present if r10Mask = L @sic R5s110178 Baseline Moving sic@  

};  }


4.3.1.1.1.1 Change 2.1.1.1.1 – Add new template cs_PktUplinkAss_IARO_SingleBlock
	Template name
	Template cs_PktUplinkAss_IARO_SingleBlock

	Reason for change
	Require Immediate Assignment RO for single block allocation

	Summary of change
	Create new template for Immediate Assignment RO for single block allocation

	TTCN module
	CommonGERAN\GERAN_Templates.ttcn

	MCC160 Comment
	


New Template
	  // Anritsu_13322 - add new template for single block assignment

  template (value) PacketUplinkAssign cs_PktUplinkAss_IARO_SingleBlock(B5_Type p_Gamma,

                                                                       B16_Type p_StartTime) :=

  { /* @status    APPROVED */

    allocMask       := '0'B,                         // 0 -> indicating single block allocation

    tFI_Assignment  := omit,                         // temporary flow identifier - not present if allocMask = 0

                                                     // not present if allocMask = 0

    pOLLING         := omit,                         // not present if allocMask = 0

    spareBit        := omit,                         // not present if allocMask = 0

    uSF             := omit,                         // not present if allocMask = 0 or fixedAllocMask = 1

    uSFGranularity  := omit,                         // not present if allocMask = 0 or fixedAllocMask = 1

    p0Mask          := omit,                         // not present if allocMask = 0 or fixedAllocMask = 1

    p0              := omit,                         // not present if allocMask = 0 or fixedAllocMask = 1 or p0Mask = 0

    pR_MODE         := omit,                         // not present if allocMask = 0 or fixedAllocMask = 1 or p0Mask = 0

    ChannelCodingCommand := omit,                    // not present if allocMask = 0

    tLLI_BlockChannelCodeing := omit,                // not present if allocMask = 0

    alphaMask       := omit,                         // not present if allocMask = 0

    aLPHA           := omit,                         // not present if allocMask = 0 or alphaMask = 0

    gMMA            := omit,                         // not present if allocMask = 0

    taMask          := omit,                         // not present if allocMask = 0

    tIMING_ADVANCE_INDEX := omit,                    // not present if allocMask = 0 or taMask = 0

    startingTimeMask := omit,                        // not present if allocMask =  0

    tBF_STARTING_TIME := omit,                       // not present if allocMask = 0 or startingTimeMask = 0

    alpha1Mask      := '0'B,                         // present if allocMask = 0 single block allocation

    aLPHA1          := omit,                         // not present if allocMask = 1 or alpha1Mask = 0

    gMMA1           := p_Gamma,                      // not present if allocMask = 1

    prefixBits      := '01'B,                        // not present if allocMask = 1 the value is 01 if present

    tBF_STARTING_TIME1 := p_StartTime,               // not present if allocMask = 1

    p02Mask         := '0'B,                         // not present if allocMask = 1

    p02             := omit,                         // not present if allocMask = 1 or p02Mask = L

    spareBit2       := omit,                         // not present if allocMask = 1 or p02Mask = L

    pR_MODE2        := omit,                         // not present if allocMask = 1 or p02Mask = L

    r99Mask         := '0'B,

    extRAMask       := omit,                         // not present if r99Mask = L

    extRA           := omit,                         // not present if r99Mask = L or extRAMask = 0

    r6Mask          := omit,                         // not present if r99Mask = L

    pfiMask         := omit,                         // not present if r99Mask = L or r6Mask = L

    pFI             := omit                          // not present if r99Mask = L, r6Mask = L or pfiMask = 0

  }; }


4.3.1.2 Change 2.1.1.2 – Add new template cs_PacketUplinkAssignment_13322
	Template name
	cs_PacketUplinkAssignment_13322

	Reason for change
	Require new template Packet Uplink Assigment 

	Summary of change
	Create new template Packet Uplink Assigment.

	TTCN module
	CommonGERAN\GERAN_Templates.ttcn

	MCC160 Comment
	


Nested Function

	  // Anritsu_13322 - new template

  template (value) GRLCCntrlMsgReqType cs_PacketUplinkAssignment_13322(B10_Type p_ARFCN, TLLI p_TLLI ) :=

  { /* This message is sent on the PCCCH or PACCH by the network to the mobile station to assign uplink resources for One phase TBF.

       TS 3GPP 44.060 clause 11.2.29

       Values are taken from 51010 clause 40.2.3.4 */

    /* @status    APPROVED */

    pktULAssignmt := {

      msgType         := '001010'B,                    // Message type

      pAGE_MODE       := int2bit(tsc_NormalPaging,2),  // Normal paging

      persistenceLevelMask := '0'B,

      persistenceLevel1 := omit,                       // not Present if persistenceLevelMask=0

      persistenceLevel2 := omit,                       // not Present if persistenceLevelMask=0

      persistenceLevel3 := omit,                       // not Present if persistenceLevelMask=0

      persistenceLevel4 := omit,                       // not Present if persistenceLevelMask=0

      globalTFIMask := '1'B,

      globalTFI := omit,

      tLLIMask := '0'B,

      tLLI := p_TLLI,

      tQIMask := omit,

      tQI := omit,

      referenceAddress := omit,

      eGPRSmask      := '0'B,

      channelCodingCommand := '00'B,                   // not present of EGPRPSmask=1

      tLLIBlockChannelCoding := '1'B,                  // 0 - the mobile station shall use CS-1 in GPRS TBF mode and MCS-1 in EGPRS TBF mode.

                                                       // 1 - the mobile station shall use the value commanded in the CHANNEL_CODING_COMMAND or EGPRS_CHANNEL_CODING_COMMAND field.

      packetTimingAdvance := cs_PacketTimingAdvance,

      freqParamMask   := '1'B,

      freqParam       := cs_FreqParam_PDTCH(p_ARFCN),   // not present if freqParamMask=0

      allocationMask  := '01'B,

      dynamicAllocation := cs_DynamicAllocationPacket, // present if allocationMask=01

      singleBlockAllocation := omit,                   // present if allocationMask=10

      r99Mask         := '0'B,

      packetExtTAmask := omit,                         // not present if r99Mask=0

      packetExtendedTA := omit,                        // not present if r99Mask=0 and packetExtTImask=0

      r5Mask          := omit,

      gRNTIMask       := omit,                         // not present if r99Mask=0 and r5Mask=0

      gRNTIExtension  := omit,                         // not present if r99Mask=0, r5Mask=0 and gRNTIMask=0

      rBIdMask        := omit,                         // not present if r99Mask=0 and r5Mask=0

      rBId            := omit,                         // not present if r99Mask=0, r5Mask=0 and rBIdMask=0

      r6Mask          := omit,                         // not present if r99Mask=0, r5Mask=0

      pfiMask         := omit,                         // not present if r99Mask=0, r5Mask=0 and r6Mask = 0

      pFI             := omit,                         // not present if r99Mask=0, r5Mask=0, r6Mask = 0 and pfiMask = 0

      rlcModeMask     := omit,                         // not present if r99Mask=0, r5Mask=0 and r6Mask = 0

      rlcMode         := omit,                         // not present if r99Mask=0, r5Mask=0, r6Mask = 0 and rlcModeMask = 0

      mask            := omit,

      eGPRSmessage    := omit,                         // not present of EGPRPSmask=0

      dualCarrier     := omit

    }

  };


5 Execution Log Files

Nvidia UE
Nvidia Icera 410 UE passed this test case on the Anritsu ME7832L LTE system with NAS and AS Snow 3G integrity and ciphering algorithm. The documentation below is enclosed as evidence of the successful test case run [1]
· Test Case Execution log file :

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file
6 References

	[1]
	  R5s120571 This archive comprises text format execution log file.

