
Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iWD-EUTRA-B2012-03_D12wk24’.

Corrections 

Change 1 – Correction to function ‘f_TC_13_1_1_EUTRA()’
	Function name
	f_TC_13_1_1_EUTRA()

	Reason for change
	As per the assumption in TS 36.523-3 clause 7.14.2, on dedicated bearers for loopback mode B, UDP packets used shall match the packet filters configured at the UE side. The UDP packets, having no specific content, shall have the correct header checksum and UDP checksum. On the default bearer, any other packets can be used, as an example, ICMPv4 ECHO REPLY similar as for loopback mode A.
However, there are UE in the market which are configured as IPv6-only UE and do not accept IPv4 packets – not even on the default bearer. Such behaviour seems to be compliant as Core Spec TS 36.509 does not specify any limitation w.r.t. PDN types to be supported for loopback mode B (see clause 5.1). Which means that the TTCN has to support all different implementation flavours in loopback mode B including IPv6-only UE which require that the to be looped-back data is a valid IPv6 packet addressing the UE (i.e. the destination address is the UE’s IPv6 address). 

However, TTCN was only distinguishing between PDN types for loopback mode B on dedicated bearers. On the default bearer, TTCN was sending an IPv4 instead of an IPv6 packet and therefore a UE not supporting IPv4 in loopback mode B  was discarding the invalid DL packet and thus failed for example TC 13.1.1 in step 6. 

	Summary of change
	Distinguishing between PDN types is now done before performing any loopback of packets, regardless of whether on dedicated or on default bearers.

Also, the name of the IPpacket has been changed to v_IPPacket to match the naming in the prose spec.

	TTCN module
	MultiLayer_Procedures.ttcn

	MCC160 Comment
	


Before change

	...

  function f_TC_13_1_1_EUTRA() runs on EUTRA_PTC
  { /* Activation and deactivation of additional data radio bearer in E-UTRA */
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var template(value) NAS_MSG_Request_Type v_NAS_MSG_Request;
    var template (value) ESM_Cause v_EsmCause36 := cs_ESM_Cause_v ('00100100'B); // "Regular deactivation"  // @sic R5-101118: ESM_Cause sic@
    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;
    var EPS_BearerIdentity v_DedicatedEpsBeareId := tsc_EpsDedicatedBearerId;
    var EPS_BearerIdentity v_LinkedEpsBearerId := tsc_EpsDefaultBearerId;
    var UInt16_Type v_SourcePort := 31160; // @sic R5s100068 sic@ // @sic R5s100135 sic@
    var UInt16_Type v_DestPort := 61000; // @sic R5s100068 sic@ // @sic R5s100135 sic@
    var template (value) octetstring v_AnyUdpPacket;
    var B3_Type v_PdnType;
    f_EUTRA_Init ( c1 );
    v_NAS_MSG_Request := cs_NAS_Request( tsc_SHT_IntegrityProtected_Ciphered,
                                         cs_508_DeactivateEPSBearerCxtReq( v_DedicatedEpsBeareId,
                                                                           v_EPS_TI,
                                                                           v_EsmCause36 ) ); // @sic R5-101118: ESM_Cause sic@
    //Create and configure cell
    f_EUTRA_CellConfig_DRB_Common ( eutra_Cell1, 1, 0);
    /* Preamble The UE is in state Generic RB Established, UE test mode activated (state 3A) according to [18] using the UE TEST LOOP MODE B. */
    f_EUTRA_Preamble ( eutra_Cell1, STATE2A_TESTLOOP_ModeB );
    f_EUTRA_RbEst_Def ( eutra_Cell1 );
    f_EUTRA_TestBody_Set ( true, layer2 );  // @sic R5-101050 sic@
    //@siclog "Step 1 - 2" siclog@
    /* Step 1: SS transmits an RRCConnectionReconfiguration message to establish a data radio bearer.
     *         The SS transmits a ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST message activating a new EPS bearer context.
     *         The NAS message is included in a RRCConnectionReconfiguration message.
     * Step 2: The UE transmit a RRCConnectionReconfigurationComplete message to confirm the establishment of data radio bearer. */
    f_EUTRA_TxDedicatedEpsBearerCtxtReq ( eutra_Cell1,
                                          tsc_DRB2,
                                          v_DedicatedEpsBeareId,
                                          v_EPS_TI,
                                          cs_LinkedId(hex2bit( v_LinkedEpsBearerId)),
                                          cs_508_EPS_QoS_Dedicated_1, // According to Context#1 in TS 36.508
                                          f_EUTRA_Get_508_TrafficFlowTemplate(1) ); // According to Context#1 in TS 36.508
    //@siclog "Step 3" siclog@
    /* The UE transmits an ACTIVATE DEDICATE EPS BEARER CONTEXT ACCEPT message. */
    SRB.receive ( car_SRB2_NasPdu_IND( eutra_Cell1,
                                       cr_NAS_Indication( tsc_SHT_IntegrityProtected_Ciphered,
                                                          cr_508_ActDedicatedEPSBearerCxtAccept ( v_DedicatedEpsBeareId ) ) ) );
    //@siclog "Step 4" siclog@
    /* The SS closes the UE test loop mode. */
    f_EUTRA_CloseUE_TestLoopModeB ( eutra_Cell1, '00'O );
    //@siclog "Step 5" siclog@
    /* The SS transmits one IP packet to the UE on the DRB associated with the default EPS bearer context. */
    DRB.send ( cas_DRB_COMMON_REQ_PDCP_SDU ( eutra_Cell1,
                                             tsc_RbId_DRB1,
                                             cs_TimingInfo_Now,
                                             { crs_IPv4_AnyIcmpEchoReply_1 ( px_IPv4_Address1_UE, // @sic R5-113734 sic@
                                                                             px_IPv4_Address1_UE ) } ) ); // @sic R5s100135, R5-113734 sic@
    //@siclog "Step 6" siclog@
    /* Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context? */
    DRB.receive ( car_DRB_COMMON_IND_PDCP_SDUList ( eutra_Cell1,
                                                    tsc_RbId_DRB1,
                                                    ?,
                                                    { crs_IPv4_AnyIcmpEchoReply_1 ( px_IPv4_Address1_UE, // @sic R5-113734 sic@
                                                                                    px_IPv4_Address1_UE ) } ) ); // @sic R5s100135 sic@
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.1.1 Step 6");
    //Check PDN type and select IP version accordingly
    v_PdnType := f_EUTRA_MobileInfo_GetAssignedPdnType();
    if (v_PdnType == tsc_PdnType_IPv4) {  /* @sic R5-112600: clarification to conditions for IP address configuration sic@ */
      v_AnyUdpPacket := crs_IPv4_AnyUdpPacket(px_IPv4_Address1_NW, px_IPv4_Address1_NW, v_SourcePort, v_DestPort); // @sic R5-113734 sic@
    } else {
      v_AnyUdpPacket := crs_IPv6_AnyUdpPacket(px_IPv6_Address1_NW, px_IPv6_Address1_NW, v_SourcePort, v_DestPort); // @sic R5-113734 sic@
    }
    //@siclog "Step 7" siclog@
    /* The SS transmits one IP packet to the UE on the DRB associated with the dedicated EPS bearer context. */
    DRB.send ( cas_DRB_COMMON_REQ_PDCP_SDU ( eutra_Cell1,
                                             tsc_RbId_DRB2,
                                             cs_TimingInfo_Now,
                                             { v_AnyUdpPacket } ) ); // @sic R5s100135 sic@
    //@siclog "Step 8" siclog@
    /* Check: Does the UE loop back the IP packet on the DRB associated with the dedicated EPS bearer context? */
    DRB.receive ( car_DRB_COMMON_IND_PDCP_SDUList ( eutra_Cell1,
                                                    tsc_RbId_DRB2,
                                                    ?,
                                                    { v_AnyUdpPacket } ) ); // @sic R5s100135 sic@
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.1.1 Step 8");
    //@siclog "Step 9" siclog@
    /*         SS transmits an RRCConnectionReconfiguration message to release a data radio bearer.
     *         The SS transmits a DEACTIVATE EPS BEARER CONTEXT REQUEST including the EPS bearer identity of the dedicated EPS bearer.
     *         The NAS message is included in a RRCConnectionReconfiguration message. */
    SRB.send ( cas_SRB1_RrcNasPdu_REQ ( eutra_Cell1,
                                        cs_TimingInfo_Now,
                                        cs_RRCConnectionReconfiguration_DRB_Release ( v_RRC_TI, tsc_DRB2 ),
                                        v_NAS_MSG_Request ) );
    //@siclog "Step 10" siclog@
    /* The UE transmits an RRCConnectionReconfigurationComplete message. */
    SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete ( v_RRC_TI ) ) );
    //@siclog "Step 11" siclog@
    /* The UE transmit a DEACTIVATE EPS BEARER CONTEXT ACCEPT. */
    SRB.receive ( car_SRB2_NasPdu_IND ( eutra_Cell1,
                                        cr_NAS_Indication ( tsc_SHT_IntegrityProtected_Ciphered,
                                                            cr_508_DeactivateEPSBearerCxtAccept ( v_DedicatedEpsBeareId ) ) ) );
    //@siclog "Step 12" siclog@
    /* The SS transmits one IP packet to the UE on the DRB associated with the default EPS bearer context. */
    DRB.send ( cas_DRB_COMMON_REQ_PDCP_SDU ( eutra_Cell1,
                                             tsc_RbId_DRB1,
                                             cs_TimingInfo_Now,
                                             { crs_IPv4_AnyIcmpEchoReply_1 ( px_IPv4_Address1_UE, // @sic R5-113734 sic@
                                                                             px_IPv4_Address1_UE ) } ) ); // @sic R5s100135, R5-113734 sic@
    //@siclog "Step 13" siclog@
    /* Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context? */
    DRB.receive ( car_DRB_COMMON_IND_PDCP_SDUList ( eutra_Cell1,
                                                    tsc_RbId_DRB1,
                                                    ?,
                                                    { crs_IPv4_AnyIcmpEchoReply_1 ( px_IPv4_Address1_UE, // @sic R5-113734 sic@
                                                                                    px_IPv4_Address1_UE ) } ) ); // @sic R5s100135, R5-113734 sic@
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.1.1 Step 13");
    f_EUTRA_TestBody_Set ( false );
    /* Postamble */
    f_EUTRA_Postamble ( eutra_Cell1, E3_TESTMODE ); //@sic R5-103875 sic@
  }
  /*
   * @desc      REFERENCE TS 36.523-1 clause 13.1.2
   * @status    APPROVED
   */
...


After change

	...

  function f_TC_13_1_1_EUTRA() runs on EUTRA_PTC
  { /* Activation and deactivation of additional data radio bearer in E-UTRA */
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var template(value) NAS_MSG_Request_Type v_NAS_MSG_Request;
    var template (value) ESM_Cause v_EsmCause36 := cs_ESM_Cause_v ('00100100'B); // "Regular deactivation"  // @sic R5-101118: ESM_Cause sic@
    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;
    var EPS_BearerIdentity v_DedicatedEpsBeareId := tsc_EpsDedicatedBearerId;
    var EPS_BearerIdentity v_LinkedEpsBearerId := tsc_EpsDefaultBearerId;
    // var UInt16_Type v_SourcePort := 31160; // @sic R5s100068 sic@ // @sic R5s100135 sic@
    // var UInt16_Type v_DestPort := 61000; // @sic R5s100068 sic@ // @sic R5s100135 sic@
    var template (value) octetstring v_IPPacketDefDrb;
    var template (value) octetstring v_IPPacketDedicatedDrb;
    var B3_Type v_PdnType;
    f_EUTRA_Init ( c1 );
    v_NAS_MSG_Request := cs_NAS_Request( tsc_SHT_IntegrityProtected_Ciphered,
                                         cs_508_DeactivateEPSBearerCxtReq( v_DedicatedEpsBeareId,
                                                                           v_EPS_TI,
                                                                           v_EsmCause36 ) ); // @sic R5-101118: ESM_Cause sic@
    //Create and configure cell
    f_EUTRA_CellConfig_DRB_Common ( eutra_Cell1, 1, 0);
    /* Preamble The UE is in state Generic RB Established, UE test mode activated (state 3A) according to [18] using the UE TEST LOOP MODE B. */
    f_EUTRA_Preamble ( eutra_Cell1, STATE2A_TESTLOOP_ModeB );
    f_EUTRA_RbEst_Def ( eutra_Cell1 );
    f_EUTRA_TestBody_Set ( true, layer2 );  // @sic R5-101050 sic@
    //@siclog "Step 1 - 2" siclog@
    /* Step 1: SS transmits an RRCConnectionReconfiguration message to establish a data radio bearer.
     *         The SS transmits a ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST message activating a new EPS bearer context.
     *         The NAS message is included in a RRCConnectionReconfiguration message.
     * Step 2: The UE transmit a RRCConnectionReconfigurationComplete message to confirm the establishment of data radio bearer. */
    f_EUTRA_TxDedicatedEpsBearerCtxtReq ( eutra_Cell1,
                                          tsc_DRB2,
                                          v_DedicatedEpsBeareId,
                                          v_EPS_TI,
                                          cs_LinkedId(hex2bit( v_LinkedEpsBearerId)),
                                          cs_508_EPS_QoS_Dedicated_1, // According to Context#1 in TS 36.508
                                          f_EUTRA_Get_508_TrafficFlowTemplate(1) ); // According to Context#1 in TS 36.508
    //@siclog "Step 3" siclog@
    /* The UE transmits an ACTIVATE DEDICATE EPS BEARER CONTEXT ACCEPT message. */
    SRB.receive ( car_SRB2_NasPdu_IND( eutra_Cell1,
                                       cr_NAS_Indication( tsc_SHT_IntegrityProtected_Ciphered,
                                                          cr_508_ActDedicatedEPSBearerCxtAccept ( v_DedicatedEpsBeareId ) ) ) );
    //@siclog "Step 4" siclog@
    /* The SS closes the UE test loop mode. */
    f_EUTRA_CloseUE_TestLoopModeB ( eutra_Cell1, '00'O );
    //Check PDN type and select IP version accordingly
    v_PdnType := f_EUTRA_MobileInfo_GetAssignedPdnType();
    // Set IP packet to be used for loopback    
    if (v_PdnType == tsc_PdnType_IPv4) {  
      v_IPPacketDefDrb := crs_IPv4_AnyUdpPacket(px_IPv4_Address1_NW, px_IPv4_Address1_NW, 5060, 5060); // NOT matching TFT
      v_IPPacketDedicatedDrb := crs_IPv4_AnyUdpPacket(px_IPv4_Address1_NW, px_IPv4_Address1_NW, 31160, 61000); // matching TFT
    } else {
      v_IPPacketDefDrb := crs_IPv6_AnyUdpPacket(px_IPv6_Address1_NW, px_IPv6_Address1_NW, 5060, 5060); 
      v_IPPacketDedicatedDrb := crs_IPv6_AnyUdpPacket(px_IPv6_Address1_NW, px_IPv6_Address1_NW, 31160, 61000); 
    }
    //@siclog "Step 5" siclog@
    /* The SS transmits one IP packet to the UE on the DRB associated with the default EPS bearer context. */
    DRB.send ( cas_DRB_COMMON_REQ_PDCP_SDU ( eutra_Cell1,
                                             tsc_RbId_DRB1,
                                             cs_TimingInfo_Now,
                                             { v_IPPacketDefDrb } ) );  // R5s120470
    //@siclog "Step 6" siclog@
    /* Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context? */
    DRB.receive ( car_DRB_COMMON_IND_PDCP_SDUList ( eutra_Cell1,
                                                    tsc_RbId_DRB1,
                                                    ?,
                                                    { v_IPPacketDefDrb } ) ); // R5s120470
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.1.1 Step 6");
    //@siclog "Step 7" siclog@
    /* The SS transmits one IP packet to the UE on the DRB associated with the dedicated EPS bearer context. */
    DRB.send ( cas_DRB_COMMON_REQ_PDCP_SDU ( eutra_Cell1,
                                             tsc_RbId_DRB2,
                                             cs_TimingInfo_Now,
                                             { v_IPPacketDedicatedDrb } ) ); // R5s120470
    //@siclog "Step 8" siclog@
    /* Check: Does the UE loop back the IP packet on the DRB associated with the dedicated EPS bearer context? */
    DRB.receive ( car_DRB_COMMON_IND_PDCP_SDUList ( eutra_Cell1,
                                                    tsc_RbId_DRB2,
                                                    ?,
                                                    { v_IPPacketDedicatedDrb } ) ); // R5s120470
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.1.1 Step 8");
    //@siclog "Step 9" siclog@
    /*         SS transmits an RRCConnectionReconfiguration message to release a data radio bearer.
     *         The SS transmits a DEACTIVATE EPS BEARER CONTEXT REQUEST including the EPS bearer identity of the dedicated EPS bearer.
     *         The NAS message is included in a RRCConnectionReconfiguration message. */
    SRB.send ( cas_SRB1_RrcNasPdu_REQ ( eutra_Cell1,
                                        cs_TimingInfo_Now,
                                        cs_RRCConnectionReconfiguration_DRB_Release ( v_RRC_TI, tsc_DRB2 ),
                                        v_NAS_MSG_Request ) );
    //@siclog "Step 10" siclog@
    /* The UE transmits an RRCConnectionReconfigurationComplete message. */
    SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete ( v_RRC_TI ) ) );
    //@siclog "Step 11" siclog@
    /* The UE transmit a DEACTIVATE EPS BEARER CONTEXT ACCEPT. */
    SRB.receive ( car_SRB2_NasPdu_IND ( eutra_Cell1,
                                        cr_NAS_Indication ( tsc_SHT_IntegrityProtected_Ciphered,
                                                            cr_508_DeactivateEPSBearerCxtAccept ( v_DedicatedEpsBeareId ) ) ) );
    //@siclog "Step 12" siclog@
    /* The SS transmits one IP packet to the UE on the DRB associated with the default EPS bearer context. */
    DRB.send ( cas_DRB_COMMON_REQ_PDCP_SDU ( eutra_Cell1,
                                             tsc_RbId_DRB1,
                                             cs_TimingInfo_Now,
                                             { v_IPPacketDefDrb } ) ); // R5s120470
    //@siclog "Step 13" siclog@
    /* Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context? */
    DRB.receive ( car_DRB_COMMON_IND_PDCP_SDUList ( eutra_Cell1,
                                                    tsc_RbId_DRB1,
                                                    ?,
                                                    { v_IPPacketDefDrb } ) ); // R5s120470
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.1.1 Step 13");
    f_EUTRA_TestBody_Set ( false );
    /* Postamble */
    f_EUTRA_Postamble ( eutra_Cell1, E3_TESTMODE ); //@sic R5-103875 sic@
  }
  /*
...


Change 2 – Correction to function ‘f_TC_13_1_5_EUTRA()’
	Function name
	f_TC_13_1_5_EUTRA()

	Reason for change
	See change 1 above. 

	Summary of change
	See change 1 above.

	TTCN module
	MultiLayer_Procedures.ttcn

	MCC160 Comment
	


Before change

	...

  function f_TC_13_1_5_EUTRA() runs on EUTRA_PTC
  {
    var GutiParameters_Type v_GutiParams;
    var IRAT_CoOrd_SysInfo_Type v_CoordInfo;
    var CoOrd_UTRANSysInfoFDD_Type v_UTRANSysInfoFDD;
    var CoOrd_UTRANSysInfoTDD_Type v_UTRANSysInfoTDD;
    var NasCount_Type v_NasCountDL;
    var octetstring v_HandoverToUtranCommand;
    var octetstring v_Nas_SecurityParamFromEUTRA;
    f_EUTRA_Init (c4);
    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_Cell1);
    //Synchronise with UTRAN PTC and get Cell Parameters for UTRAN cell
    v_CoordInfo := f_IRAT_WaitForCoOrd_SysInfo (UTRAN);
    //Build the HO-to-UTRAN command
    if (ischosen (v_CoordInfo.Utran[0].FDD)) {
      v_UTRANSysInfoFDD := v_CoordInfo.Utran[0].FDD;
      v_HandoverToUtranCommand := f_U_EncodeHandoverToUtran_64kPS_DCH4_FDD(v_UTRANSysInfoFDD.UplinkScrambingCode,
                                                                           v_UTRANSysInfoFDD.ScramblingCode,
                                                                           v_UTRANSysInfoFDD.Freq);
    }
    else {
      v_UTRANSysInfoTDD := v_CoordInfo.Utran[0].TDD;
      v_HandoverToUtranCommand := f_U_EncodeHandoverToUtran_64kPS_DCH4_TDD(v_UTRANSysInfoTDD.CellParametersID,
                                                                           v_UTRANSysInfoTDD.Freq);
    }
    //Create and configure cell
    f_EUTRA_CellConfig_Def (eutra_Cell1);
    f_IRAT_WaitForCoOrd_Trigger(UTRAN); // wait for UTRAN side to do pre-preamble @sic R5s110776 sic@
    //Bring UE to initial state
    f_EUTRA_Preamble ( eutra_Cell1, STATE2A_TESTLOOP_ModeB);
    f_EUTRA_RbEst_Def ( eutra_Cell1 );
    f_EUTRA_CloseUE_TestLoopModeB ( eutra_Cell1, '00'O );
    //Tell UTRA PTC to set appropiate cell levels
    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);
    f_EUTRA_TestBody_Set ( true, layer2 );
    //@siclog "Step 1" siclog@
    //The SS transmits one IP packet to the UE on the DRB associated with the default EPS bearer context
    DRB.send (cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1,
                                          tsc_RbId_DRB1,
                                          cs_TimingInfo_Now,
                                          { crs_IPv4_AnyIcmpEchoReply_1 ( px_IPv4_Address1_UE, // @sic R5-113734 sic@
                                                                          px_IPv4_Address1_UE )})); // @sic R5-113734 sic@
    //@siclog "Step 2" siclog@
    //Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context?
    DRB.receive (car_DRB_COMMON_IND_PDCP_SDUList(eutra_Cell1,
                                                 tsc_RbId_DRB1,
                                                 ?,
                                                 { crs_IPv4_AnyIcmpEchoReply_1 ( px_IPv4_Address1_UE, // @sic R5-113734 sic@
                                                                                 px_IPv4_Address1_UE )})); // @sic R5-113734 sic@
    //@siclog "Step 3" siclog@
    //Make the UE initiate an outgoing call
    f_UT_RequestCSCall(UT);
    //@siclog "Step 4" siclog@
    //The UE transmits an ULInformationTransfer message including Extanded Service Request
    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_EXT_SERVICE_REQUEST_MO(f_EUTRA_SecurityKSIasme_Get(),
                                                                                bit2oct(v_GutiParams.M_TMSI)))));
    //@siclog "Step 4A - 4B" siclog@
    f_EUTRA_UTRA_Capability (eutra_Cell1); //@sic R5-120776 sic@
    //Get NAS DL Count
    v_NasCountDL := f_EUTRA_SS_NAS_CountGetDLNext(); // @sic R5s110382 sic@
    //nas-securityParamFromEUTRA is the 4 least significant bits of the NAS downlink COUNT value 36.508, Table 4.6.1-6
    v_Nas_SecurityParamFromEUTRA := substr( v_NasCountDL, 3, 1);
    //Wait till SS local configuration is complete
    f_IRAT_WaitForCoOrd_Trigger (UTRAN);
    //@siclog "Step 5" siclog@
    //Stop periodic TA on source cell...
    //@sic R5s120342 sic@
    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_PUCCH_Synch_Config_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_PUCCH_Synch_None));
    //The SS transmits a MobilityFromEUTRACommand message on Cell 1
    SRB.send ( cas_SRB1_RrcPdu_REQ ( eutra_Cell1,
                                     cs_TimingInfo_Now,
                                     cs_MobilityFromEutraCommand_Handover_UTRAN_CSFB ( v_HandoverToUtranCommand,
                                                                   v_Nas_SecurityParamFromEUTRA ) ) );
    //Wait till test case ends
    f_IRAT_WaitForCoOrd_Trigger (UTRAN);
    f_EUTRA_TestBody_Set(false);
    //Release non serving cell(s)
    f_EUTRA_ReleaseAllCells ();
  }
...


After change

	...

  function f_TC_13_1_5_EUTRA() runs on EUTRA_PTC
  {
    var GutiParameters_Type v_GutiParams;
    var IRAT_CoOrd_SysInfo_Type v_CoordInfo;
    var CoOrd_UTRANSysInfoFDD_Type v_UTRANSysInfoFDD;
    var CoOrd_UTRANSysInfoTDD_Type v_UTRANSysInfoTDD;
    var NasCount_Type v_NasCountDL;
    var octetstring v_HandoverToUtranCommand;
    var octetstring v_Nas_SecurityParamFromEUTRA;
    var template (value) octetstring v_IPPacket;
    var B3_Type v_PdnType;
    var UInt16_Type v_SourcePort := 31160; 
    var UInt16_Type v_DestPort := 61000; 
    f_EUTRA_Init (c4);
    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_Cell1);
    //Synchronise with UTRAN PTC and get Cell Parameters for UTRAN cell
    v_CoordInfo := f_IRAT_WaitForCoOrd_SysInfo (UTRAN);
    //Build the HO-to-UTRAN command
    if (ischosen (v_CoordInfo.Utran[0].FDD)) {
      v_UTRANSysInfoFDD := v_CoordInfo.Utran[0].FDD;
      v_HandoverToUtranCommand := f_U_EncodeHandoverToUtran_64kPS_DCH4_FDD(v_UTRANSysInfoFDD.UplinkScrambingCode,
                                                                           v_UTRANSysInfoFDD.ScramblingCode,
                                                                           v_UTRANSysInfoFDD.Freq);
    }
    else {
      v_UTRANSysInfoTDD := v_CoordInfo.Utran[0].TDD;
      v_HandoverToUtranCommand := f_U_EncodeHandoverToUtran_64kPS_DCH4_TDD(v_UTRANSysInfoTDD.CellParametersID,
                                                                           v_UTRANSysInfoTDD.Freq);
    }
    //Create and configure cell
    f_EUTRA_CellConfig_Def (eutra_Cell1);
    f_IRAT_WaitForCoOrd_Trigger(UTRAN); // wait for UTRAN side to do pre-preamble @sic R5s110776 sic@
    //Bring UE to initial state
    f_EUTRA_Preamble ( eutra_Cell1, STATE2A_TESTLOOP_ModeB);
    f_EUTRA_RbEst_Def ( eutra_Cell1 );
    f_EUTRA_CloseUE_TestLoopModeB ( eutra_Cell1, '00'O );
    //Tell UTRA PTC to set appropiate cell levels
    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);
    f_EUTRA_TestBody_Set ( true, layer2 );
     v_PdnType := f_EUTRA_MobileInfo_GetAssignedPdnType();
    // Set IP packet to be used for loopback    

if (v_PdnType == tsc_PdnType_IPv4) { 
      v_IPPacket := crs_IPv4_AnyUdpPacket(px_IPv4_Address1_NW, px_IPv4_Address1_NW, v_SourcePort, v_DestPort); 
    } else {
      v_IPPacket := crs_IPv6_AnyUdpPacket(px_IPv6_Address1_NW, px_IPv6_Address1_NW, v_SourcePort, v_DestPort); 
    }
    //@siclog "Step 1" siclog@
    //The SS transmits one IP packet to the UE on the DRB associated with the default EPS bearer context
    DRB.send (cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1,
                                          tsc_RbId_DRB1,
                                          cs_TimingInfo_Now,
                                          { v_IPPacket})); // R5s120470
    //@siclog "Step 2" siclog@
    //Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context?
    DRB.receive (car_DRB_COMMON_IND_PDCP_SDUList(eutra_Cell1,
                                                 tsc_RbId_DRB1,
                                                 ?,
                                                 { v_IPPacket})); // R5s120470
    //@siclog "Step 3" siclog@
    //Make the UE initiate an outgoing call
    f_UT_RequestCSCall(UT);
    //@siclog "Step 4" siclog@
    //The UE transmits an ULInformationTransfer message including Extanded Service Request
    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_EXT_SERVICE_REQUEST_MO(f_EUTRA_SecurityKSIasme_Get(),
                                                                                bit2oct(v_GutiParams.M_TMSI)))));
    //@siclog "Step 4A - 4B" siclog@
    f_EUTRA_UTRA_Capability (eutra_Cell1); //@sic R5-120776 sic@
    //Get NAS DL Count
    v_NasCountDL := f_EUTRA_SS_NAS_CountGetDLNext(); // @sic R5s110382 sic@
    //nas-securityParamFromEUTRA is the 4 least significant bits of the NAS downlink COUNT value 36.508, Table 4.6.1-6
    v_Nas_SecurityParamFromEUTRA := substr( v_NasCountDL, 3, 1);
    //Wait till SS local configuration is complete
    f_IRAT_WaitForCoOrd_Trigger (UTRAN);
    //@siclog "Step 5" siclog@
    //Stop periodic TA on source cell...
    //@sic R5s120342 sic@
    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_PUCCH_Synch_Config_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_PUCCH_Synch_None));
    //The SS transmits a MobilityFromEUTRACommand message on Cell 1
    SRB.send ( cas_SRB1_RrcPdu_REQ ( eutra_Cell1,
                                     cs_TimingInfo_Now,
                                     cs_MobilityFromEutraCommand_Handover_UTRAN_CSFB ( v_HandoverToUtranCommand,
                                                                   v_Nas_SecurityParamFromEUTRA ) ) );
    //Wait till test case ends
    f_IRAT_WaitForCoOrd_Trigger (UTRAN);
    f_EUTRA_TestBody_Set(false);
    //Release non serving cell(s)
    f_EUTRA_ReleaseAllCells ();
  }
...


Change 3 – Correction to other Multilayer, RRC and EMM TCs
	Function name
	Above change 2 has also to be applied to the following Multilayer Procedures TCs

- TC 13.1.10: steps1/2

- TC 13.3.1.1: steps1/5

- TC 13.3.1.2: steps1/8

- TC 13.4.1.2: steps1/2 and steps9/10

- TC 13.4.1.3: steps1/2, steps9/10 and steps15/16

- TC 13.4.2.1: steps1/2

- TC 13.4.2.2: steps1/2

- TC 13.4.2.4: steps17/18

- TC 13.4.2.5: steps17/18

and similarily also to those RRC and EMM TCs in which loopback mode B is used. 

	Reason for change
	See change 1 above.

	Summary of change
	See change 1 above.

	TTCN module
	MultiLayer_Procedures.ttcn

	MCC160 Comment
	


Before change

	...

...


After change

	...

...

